

Gröbner Bases for Everyone with CoCoA-5 and CoCoALib

John Abbott, Anna Maria Bigatti

Abstract.

We present a survey on the developments on Gröbner bases showing explicit examples in CoCoA.

The CoCoA project dates back to 1987: its aim was to create a “mathematician”-friendly laboratory for studying Commutative Algebra, most especially Gröbner bases. Since then, always maintaining this “friendly” tradition, it has evolved and has been completely rewritten.

CoCoA offers Gröbner bases for all levels of interest: from the basic quick call in the interactive system CoCoA-5 [4], to problem-specific optimized implementations, to the computer–computer communication with the open source C++ software library, CoCoALib [5], or the prototype OpenMath-based server.

The openness and clean design of CoCoALib and CoCoA-5 are intended to offer different levels of usage, and to encourage external contributions.

§1. Introduction

The CoCoA project dates back to 1987 under the lead of L. Robbiano: the aim was to create a software laboratory for studying Commutative Algebra and especially Gröbner bases, and which is welcoming even to mathematicians who are wary of new-fangled computers,

Since then the realm of applicability of Gröbner bases has continually expanded, so the researchers interested in using them now come from a broad palette of subject areas ranging from the theoretical to quite practical topics. So there are still the “pure” mathematicians as at the outset, but now also “programming” mathematicians, and statisticians, computer scientists, and so on. A factor crucial in making Gröbner

2010 *Mathematics Subject Classification.* 13P25, 13P10, 13-04, 14Q10, 68W30 .

Key words and phrases. Groebner bases, elimination, software library.

bases relevant to practical problems is the interim progress in computer hardware and software techniques.

Since its beginning the CoCoA project has evolved and has been rewritten, and now comes in the form of a very flexible software combination CoCoA-5/CoCoALib, while maintaining its tradition of being *user-friendly* so it offers Gröbner bases for all levels of interest and programming ability: *a Gröbner basis for everyone*. This means that the “CoCoA experience” covers a wide range: from the basic, quick call in the interactive system CoCoA-5 [4], to the problem-specific optimized implementations, to the computer-computer communication with the open source C++ software library, CoCoALib [5], or with the prototype OpenMath-based server.

The importance that Gröbner bases have acquired derives from the fact they enable or facilitate so many other computational mathematical results. A natural consequence is that a Gröbner basis is almost never the final answer that is sought, but just a stepping stone on the way to the goal. It is very rare that anyone really wants to have a Gröbner basis, let alone actually look at one — usually they are frighteningly large beasts!

1.1. What is new in CoCoA-5? And what is not?

CoCoA-4 was widely appreciated for its ease of use, and the naturalness of its interactive language. However, it did have limitations, and several “grey areas”. We designed the new CoCoA-5 language to strike a balance between backward-compatibility (hoping not to alienate existing CoCoA-4 users) and greater expressibility with a richer and more solid mathematical basis (eliminating those “grey areas”).

So, what’s not new? Superficially the new CoCoA-5 language/system closely resembles CoCoA-4 because we kept it largely backward compatible; at the same time it improves the naturalness and ease of use of the old system. We are very aware that a number of CoCoA users are mathematicians with only limited programming experience, for whom learning CoCoA was a “big investment”, and who are reluctant to make another big investment — that is why we wanted to make the passage to CoCoA-5 as painless as possible.

So, if almost nothing has changed, what’s new? The clearly defined semantics of the CoCoA-5 new language make it both more robust and more flexible; it provides greater expressibility and a more solid mathematical basis. In particular, it offers full flexibility for the field of coefficients: *e.g.* fraction fields and algebraic extensions, and even heuristically guaranteed floating point arithmetics with rational reconstruction (see Section 7.1).

However, under the surface, the change is radical since its mathematical core, CoCoALib, has been rewritten from scratch, to be faster, cleaner and more powerful than the old system, and also to be used as a C++ library.

1.2. How Do CoCoALib and CoCoA-5 differ?

The glib answer is: *As little as possible!*

A more honest answer is that that is our aim, but there is more work still to do. In any case, using CoCoA-5 will not magically make you into a C++ programmer, so some differences will necessarily remain.

CoCoA-5 is an interactive, interpreted environment which makes it far better suited to “rapid prototyping” than the rigid, statically typed regime of C++. To keep it simple to learn, CoCoA-5 has only a few data types: for instance, a power-product in CoCoA-5 is represented as a monic polynomial with a single term, *i.e.* a ring element (of a polynomial ring); in contrast, in CoCoALib there is a dedicated class, `PPMonoidElem`, which directly represents each power-product, and allows efficient operations on the values (*e.g.* without the overhead of the superfluous coefficients).

CoCoALib [1] contains (practically) all the mathematical knowledge and ability while CoCoA-5 offers convenient access to CoCoALib’s capabilities. Programming with CoCoALib tends to be more onerous than with CoCoA-5 largely because of C++’s demanding, rigid rules; the reward is typically faster computation (sometimes much faster). Also, of course, those who want to use CoCoALib’s abilities in their own C++ program have to use CoCoALib, see Section 8.

This is why everything which can be computed with CoCoA-5 should be just as readily computable with CoCoALib. Some CoCoA-5 functions are still implemented in CoCoA-5 packages; but they are being steadily translated into C++. (We don’t say which ones, because the list is constantly shrinking)

§2. Gröbner Bases with Ease

The simplest context for Gröbner bases is for ideals in $\mathbb{Q}[x, y, z]$ or $\mathbb{Z}/(p)[x, y, z]$ with p a prime. These are also the easiest cases to give to CoCoA.

An essential ingredient in the definition of a Gröbner basis is the term-ordering: a total ordering on the power-products which respects multiplication, and where 1 is the smallest power-product. In CoCoA the term-ordering is specified at the same time as the polynomial ring; all Gröbner bases of ideals in that polynomial ring will automatically be

computed with respect to that ordering — unless otherwise specified, the ordering is “degree reverse lexicographic”, which is known to be the “best general choice”. CoCoA has a fully general (matrix-based) implementation of term-orderings, with efficient handling of some common special cases.

Here is an example of computing a Gröbner basis in CoCoA:

```
/**/ use QQ[x,y,z];
/**/ I := ideal(x^3 +x*y^2 -2*z, x^2*y^3 -y*z^2);
/**/ GBasis(I);
[x^3 +x*y^2 -2*z, x^2*y^3 -y*z^2, x*y^5 -2*y^3*z +x*y*z^2, ...]
```

As mentioned above, the default ordering is **StdDegRevLex**, which *generally* gives the best performance and smallest answer. However, other gradings and orderings may be useful for studying specific problems: for instance, an important family are the *elimination orderings* which are used implicitly in Section 5.

Another well-known ordering is **lex**, which gives a Gröbner basis with a particular *shape* useful for solving polynomial systems (see, for example, the Kreuzer–Robbiano book [15], Sec.3.7). Their practical usefulness is limited by the fact that **lex** bases tend to be particularly big and ugly, and are frequently rather costly to compute.

In CoCoA the term-ordering is an intrinsic property of a polynomial ring: this means that $\mathbb{Q}[x, y, z]$ with **lex** is viewed as a different ring from $\mathbb{Q}[x, y, z]$ with **StdDegRevLex**. Here is an example of computing a **lex** Gröbner basis; note how the term-ordering is specified at the start along with the polynomial ring.

```
/**/ use QQ[x,y,z], lex; // specify ordering together with ring
/**/ I := ideal(x^3 +3, y-x^2, z-x-y);
/**/ ReducedGBasis(I); // basis is wrt. lex ordering
[z^3 +9*z -6,
 y -(1/4)*z^2 -(3/4)*z -3/2,
 x +(1/4)*z^2 -(1/4)*z +3/2]
```

In the last example above we used the command **ReducedGBasis** which computes a reduced Gröbner basis which is a “cleaned up” Gröbner basis with only non-redundant, monic, fully reduced elements — it is unique (up to the order of the elements).

2.1. More rings of coefficients

The easy examples above show the definition of a polynomial rings with rational coefficients, but the choice in CoCoA is quite wide. For example $\mathbb{Z}/(p)$ where p may also be quite large, or algebraic extensions, of fraction fields.

```

/**/ use ZZ/(10000000000000000000000000319)[x];
/**/ ReducedGBasis(ideal(3*x-1));
[x -3333333333333333333333333333440]

/**/ use R ::= QQ[i];
/**/ QQi := R/ideal(i^2 +1);
/**/ use QQi[x,y,z];
/**/ I := ideal(i*x^3 -z, x^2*y^3 -i*y*z^2);
/**/ ReducedGBasis(I);
[x^3 +(i)*z, y^3*z +x*y*z^2, x^2*y^3 +(-i)*y*z^2]

/**/ use R ::= QQ[sqrt2, sqrt3];
/**/ K := R/ideal(sqrt2^2 -2, sqrt3^2 -3);
/**/ use K[x,y,z];
/**/ I := ideal(sqrt3*x^2 -y, x*y -sqrt2*z);
/**/ ReducedGBasis(I);
[y^2 +(-sqrt2*sqrt3)*x*z, x*y +(-sqrt2)*z, x^2 +((-1/3)*sqrt3)*y]

/**/ use QQab ::= QQ[a,b];
/**/ K := NewFractionField(QQab);
/**/ use K[x,y,z];
/**/ I := ideal(x^3 -a*z, x^2*y^3 -b*y*z^2);
/**/ ReducedGBasis(I);
/**/ [x^3 -a*z, y^3*z +(-b/a)*x*y*z^2, x^2*y^3 -b*y*z^2]

```

Please note in this last example K is actually the field $\mathbb{Q}(a, b)$ with no specialization of $a, b \in \mathbb{Q}$. So the Gröbner basis represents the *generic* case, meaning that an algebraic expression in a, b which is not identically zero is considered to be non-zero. The problem of considering all possible specializations of the parameters is known as *comprehensive Gröbner basis*, and is not (yet) implemented in CoCoA.

Another family of computationally interesting rings in CoCoA is given by `NewRingTwinFloat(BitPrec)`. It will be presented in detail in Sections 4 and 7.1.

§3. Gröbner fan and universal Gröbner bases

There is a notion of *universal Gröbner basis* which is a Gröbner basis for every term-ordering. The CoCoA function “`UniversalGBasis`” will compute one such basis; this function is based on the computation of the Gröbner fan (a richer structure, see below) which gives all possible reduced Gröbner bases: we can take the union of all of them to produce the universal basis.

The following example shows that the maximal minors of a 3×4 matrix of indeterminates form a universal Gröbner basis of the ideal they generate:

```

/**/ use R := QQ[a,b,c,d,e,f,g,h,i,j,k,l];
/**/ I := ideal(minors(mat([a,b,c,d],[e,f,g,h],[i,j,k,l]),3));
/**/ indent(UniversalGBasis(I));
[
  d*g*j -c*h*j -d*f*k +b*h*k +c*f*l -b*g*l,
  d*g*i -c*h*i -d*e*k +a*h*k +c*e*l -a*g*l,
  d*f*i -b*h*i -d*e*j +a*h*j +b*e*l -a*f*l,
  c*f*i -b*g*i -c*e*j +a*g*j +b*e*k -a*f*k
]
/**/ EqSet(-1*gens(I), ReducedGBasis(I));
true

```

The **Gröbner fan** of an ideal was defined by Mora and Robbiano in 1988 ([18]): it is a (finite) fan of polyhedral cones indexing the reduced Gröbner bases of the ideal. This has been implemented by Jensen in his software *Gfan* ([16]) which he has recently linked into CoCoA; we note that CoCoA's fully general approach to representing term-orderings was essential in making this integration possible.

The Gröbner fan is useful because many well-known theoretical applications of Gröbner bases rely on the existence of a Gröbner basis of an ideal with prescribed properties, such as having a certain cardinality, or comprising polynomials of a specified degree, or all squarefree. For example, if an ideal $I \in K[x_1, \dots, x_n]$ has a Gröbner basis for *some* term-ordering comprising just quadrics, then the algebra $K[x_1, \dots, x_n]/I$ is Koszul.

The function **GroebnerFanIdeals(I)** returns all reduced Gröbner bases of the ideal I as a list of ideals: we chose to encode each basis in an ideal because in CoCoA an ideal I stores internally various pieces of information, including the term-ordering and the Gröbner basis, so putting the result in an ideal instead of a simple list of polynomials is more efficient for any subsequent operations.

The following ideal, Example 3.9 from Sturmfels's book [19], has 360 distinct reduced Gröbner bases:

```

/**/ use R := QQ[a,b,c];
/**/ I := ideal(a^5+b^3+c^2-1, a^2+b^2+c-1, a^6+b^5+c^3-1);
/**/ L := GroebnerFanIdeals(I);
/**/ len(L);
360

```

Storing all the possible different (reduced) Gröbner bases is practicable only for a small example; larger ideals may have thousands or even millions of different Gröbner bases. Typically we are interested only in those bases satisfying a certain property. So in CoCoA there is the function `CallOnGroebnerFanIdeals` which calls a given function on each Gröbner basis successively without having to store them all in a big list (if the computer's memory can handle it!). Using this CoCoA function needs a little technical ability, but might make the difference between getting an answer or not because the computer's memory filled up:

```
define GBOfLen3(I)
  if len(GBasis(I))=3 then
    println; println OrdMat(RingOf(I));
    println GBasis(I);
  endif;
enddefine;

/**/ use R := QQ[a,b,c];
/**/ I := ideal(a^5+b^3+c^2-1, b^2+a^2+c-1, c^3+a^6+b^5-1);
/**/ CallOnGroebnerFanIdeals(I, GBOfLen3);
matrix(ZZ,
  [[3, 7, 7],
   [3, 6, 8],
   [0, 0, -1]])
[b^2+c+a^2-1,
 a^5+c^2-b*c-a^2*b+b-1,
 c^3+b*c^2+2*a^2*b*c+a^4*b-a*c^2+a*b*c+a^3*b-2*b*c-2*a^2*b-a*b+b+a
 -1]
matrix(ZZ,
  [[6, 7, 14],
   [6, 5, 15],
   [0, 0, -1]])
[c+b^2+a^2-1,
 -b^6-3*a^2*b^4-3*a^4*b^2+b^5+3*b^4+6*a^2*b^2+3*a^4-3*b^2-3*a^2,
 a^5+b^4+2*a^2*b^2+a^4+b^3-2*b^2-2*a^2]
```

Here we see explicitly that CoCoA represents some term-orderings via matrices of integers. See Section 5 for an example of how to ask CoCoA to compute a Gröbner basis with a term-ordering given by a matrix.

§4. Leading Term Ideals and “gin”

Let $P = K[x_1, \dots, x_n]$ be a polynomial ring over a field K , and let σ be a term-ordering on the power-products in P . If I is an ideal in P then we define its **leading term ideal** with respect to σ , written $\text{LT}_\sigma(I)$, to be the ideal generated by the leading power-products of all non-zero polynomials in I ; some authors use the name “initial ideal” for this notion. A generating set for $\text{LT}_\sigma(I)$ may easily be obtained: we compute a reduced σ -Gröbner basis for I then collect the σ leading terms of the elements of the basis. Remarkably $\text{LT}_\sigma(I)$ captures some interesting “combinatorial” information about the original polynomial ideal I : for instance, its Hilbert series — so calling `HilbertSeries(R/I)` contains a “hidden” call to `GBasis(I)`.

A more sophisticated tool in Commutative Algebra is the **generic initial ideal** of a polynomial ideal I . This is useful because it encodes more geometrical properties of I into a monomial ideal. It is defined by taking a generic change of coordinates g (i.e. $g(x_j) = \sum_{i=1}^n a_{i,j}x_i$ in $K(a_{i,j})[x_1, \dots, x_n]$) then $\text{gin}_\sigma(I) = \text{LT}_\sigma(g(I))$. And here we have to admit that the acronym *gin* sounds nicer than *gLT*!

The definition of *gin* suggests an obvious algorithm for computing it (see Section 2.1 for an example with *generic* coefficients). However, it quickly becomes apparent that the coefficients in $K(a_{i,j})$ become unwieldy except for the very simplest cases; so the obvious approach is utterly hopeless. Instead we can pick an explicit, random change of coordinates h , and then compute $\text{LT}_\sigma(h(I))$; the coordinate changes for which $\text{gin}_\sigma(I) = \text{LT}_\sigma(h(I))$ form a Zariski-open set. This approach can be used when K is infinite; if the random change of coordinates is chosen from a large set then $\text{LT}_\sigma(h(I))$ will indeed be $\text{gin}_\sigma(I)$ with high probability. This is what CoCoA does.

While choosing random changes of coordinates with large coefficients increases the probability of getting the correct result, it also tends to produce large coefficients in the transformed polynomials. In this example the original polynomials have very small coefficients, but there is a coefficient with almost 50 digits in the transformed polynomials:

```

/**/ use P ::= QQ[x,y,z];
/**/ I := ideal(y^20 -x^5*z^6, x^2*z^3 -y*z^2);
/**/ L := [sum([ random(-1000,1000)*indet(P,j) | j in 1..3])
           | i in 1..3];

/**/ L;
[941*x -70*y -981*z, 974*x +342*y -789*z, 398*x +112*y -942*z]
/**/ g := PolyAlgebraHom(P, P, L);

```



```

/**/ GI := ideal(apply(g, gens(I))); GI;
ideal(101148714265738572816020149152880657871065317376*x^20+... ,
.....)

```

With coefficients like that, computing the Gröbner basis of the transformed ideal over the rationals would be quite expensive! Thus, when choosing random coefficients, one needs to strike a balance between a wide range, so the transformation is “generic enough”, but not too wide, to limit coefficient growth during the computation.

The implementation for computing `gin` in CoCoA uses a special approximate floating-point representation for rational coefficients, namely *twin-floats* (see Section 7.1). The Gröbner basis of the twin-float transformed ideal will only have approximate twin-float coefficients, but this does not matter because we need only the leading power-products of the polynomials in the basis.

Twin-float numbers have fixed-precision (so do not grow in size the way rational numbers do), and employ heuristics to guarantee the correctness of results. This allows the implementation to make random coefficient choices from a wide range (in fact, integers between -10^6 and 10^6) without paying the price for calculating with transformed polynomials having complicated rational coefficients. If the chosen precision for the twin-floats is too low, this will be signalled; and the computation can be restarted choosing a higher precision.

CoCoA’s function `gin` does this all automatically. Moreover, it tries a second random change of coordinates, just to make sure it gets the same leading term ideal. The internal workings can be seen via printed messages when using the option “**verbose**”.

```

/**/ gin(I, "verbose");
[
  15915*x,
  872152*x -383743*y,
  412211*x -406393*y -383480*z
]
-- trying with FloatPrecision 64
[
  -925894*x,
  327379*x -729412*y,
  -945709*x +550455*y +499099*z
]
-- trying with FloatPrecision 64
ideal(x^5, x^4*y^16, x^3*y^18, x^2*y^20, x*y^22, y^24)

```

§5. Elimination and related functions

Elimination is a central topic in Computational Commutative Algebra (see for example the text book by Kreuzer and Robbiano [15], Sec. 3.4) and its applications are countless. Elimination means: given an ideal $I \in K[t_1, \dots, t_a, x_1, \dots, x_b]$, find a set of generators of the ideal $I \cap K[x_1, \dots, x_b]$ where the indeterminates $\{t_1, \dots, t_a\}$ have been “eliminated”.

Given its usefulness, elimination is an operation offered in almost all Computer Algebra Systems. We can be sure that all such elimination functions internally compute a Gröbner basis with respect to an **elimination ordering** for the subset of indeterminates to be eliminated: with such an ordering the subset of polynomials in the Gröbner basis whose leading terms are not divisible by any of the t_j are exactly the generators we seek for the ideal $I \cap K[x_1, \dots, x_b]$.

In the example below we see the process we described and compare it with the actual output of CoCoA’s own function `elim`. Note that in both cases the generators are not minimal, but they are indeed a Gröbner basis of the elimination ideal (wrt. to the restriction of the elimination term-ordering used).

```

/**/ M := ElimMat(4, [1]); M;
matrix(ZZ,
  [[1, 0, 0, 0],
   [1, 1, 1, 1],
   [0, 0, 0, -1],
   [0, 0, -1, 0]])
/**/ P := NewPolyRing(QQ, "t, x,y,z", M, 0); // 0-grading
/**/ use P;
/**/ I := ideal(x-t, y-t^2, z-t^3);
/**/ GBasis(I);
[-t +x, -x^2 +y, -x*y +z, y^2 -x*z]
/**/ elim([t], I);
ideal(-x^2 +y, -x*y +z, y^2 -x*z)
/**/ MinSubsetOfGens(ideal(-x^2 +y, -x*y +z, y^2 -x*z));
[-x^2 +y, -x*y +z]

```

The simple example above indeed shows a particular application of `elim`, the **presentation of an algebra** $K[f_1, \dots, f_n] \simeq K[x_1, \dots, x_n]/I$. More precisely, let f_1, \dots, f_n be elements in $K[t_1, \dots, t_s]$, where $\{t_1, \dots, t_s\}$ is another set of indeterminates (viewed as parameters) and consider the K -algebra homomorphism

$$\phi : K[x_1, \dots, x_n] \longrightarrow K(t_1, \dots, t_s) \text{ given by } x_i \mapsto f_i \text{ for } i = 1, \dots, n$$

Its kernel is a prime ideal, and the general problem of **implicitization** is to find a set of generators for this ideal.

The Gröbner basis elimination technique consists of defining the ideal $J = \langle x_1 - f_1, \dots, x_n - f_n \rangle$ in the ring $K[t_1, \dots, t_s, x_1, \dots, x_n]$ and *eliminating* all the parameters t_i , as we saw in the example.

Unfortunately this extraordinarily elegant tool often turns out to be quite inefficient, resulting in long and costly computations. Knowing how to exploit special properties of a given class of examples might make a huge difference.

5.1. Toric

If the algebra we want to present is generated by power-products then the elimination can be computed by the CoCoA function “**toric**”; toric ideals are prime and generated by binomials.

Again we consider $\mathbb{Q}[t, t^2, t^3]$, and also $\mathbb{Q}[t^3, t^4, t^5]$:

```

/**/ use QQ[x,y,z];
/**/ toric(RowMat([1,2,3])); // just the list of exponents
ideal(-x^2 +y, x^3 -z)

/**/ use QQ[x,y,z];
/**/ toric(RowMat([3,4,5]));
ideal(-y^2 +x*z, x^3 -y*z, -x^2*y +z^2)

```

With a very slightly more challenging example we can clearly measure the advantage in using the specialized function “**toric**” over the general function “**elim**”:

```

/**/ use R ::= ZZ/(2)[x[1..6], s,t,u,v];
/**/ L := [s*u^20, s*u^30, s*t^20*v, t*v^20, s*t*u*v, s*t^2*u];
/**/ ExpL := [[ 1, 1, 1, 0, 1, 1],
               [ 0, 0, 20, 1, 1, 2],
               [20, 30, 0, 0, 1, 1],
               [ 0, 0, 1, 20, 1, 0]];

/**/ I := ideal([x[i] - L[i] | i in 1..6]);
/**/ t0 := CpuTime(); IE := elim([s,t,u,v], I); TimeFrom(t0);
9.274
/**/ t0 := CpuTime(); IT := toric(ExpL); TimeFrom(t0);
0.032

```

The CoCoA function “**toric**” employs a non-deterministic algorithm: so the actual set of ideal generators produced might vary.

For further details on the algorithms implemented in CoCoA see Bigatti, La Scala, Robbiano [13]. The article describes three different

algorithms; the default one in CoCoA is EATI (*Elimination Algorithm for Toric Ideals*).

For more details on the specific function “**toric**” type **?toric** into CoCoA (or read the PDF manual, or the html manual in the web-site).

5.2. Implicitization of hypersurfaces

As mentioned earlier elimination provides a general solution to the implicitization problem, but this solution is more elegant than practical. We can do rather better in the special case of implicitization of a hypersurface. One immediate feature is that the result is really just a single polynomial since the eliminated ideal must be principal.

It is well-known that Buchberger’s algorithm usually works better with homogeneous ideals (though there are sporadic exceptions). Yet the very construction of the eliminating ideal $J = \langle x_1 - f_1, \dots, x_n - f_n \rangle$ looks intrinsically non-homogeneous. But with a little well-guided effort we can transform the problem into the calculation of a Gröbner basis of a homogeneous ideal.

If the f_j are all polynomials then we take a new indeterminate (say h) and use it to homogenize each f_j to produce F_j . Now we do have a homogeneous ideal $J' = \langle x_1 - F_1, \dots, x_n - F_n \rangle$ provided we give weights to the x_i indeterminates by setting $\deg(x_i) = \deg(f_i)$ for each i .

Since we are in the special case of a hypersurface, it can be shown that the (non-zero) polynomial of lowest degree in J' is unique up to scalar multiples; its dehomogenization is then the polynomial we seek! We get two advantages from the homogeneous ideal J' : we gain efficiency by using Buchberger’s algorithm degree-by-degree, and we can stop as soon as the first basis polynomial is found — most probably there will still be many pairs to process. See Abbott, Bigatti, Robbiano [7] for all details and proofs, and also how to “correctly homogenize” parametrizations by rational functions.

```

/**/ use P ::= QQ[s,t, x,y,z];
/**/ elim([s,t], ideal(x-s^2, y-s*t, z-t^2) );
ideal(y^2 -x*z)

/**/ use R ::= QQ[s,t];
/**/ P ::= QQ[x,y,z];
/**/ ImplicitHypersurface(P, [s^2, s*t, t^2], "ElimTH");
ideal(y^2 -x*z)

```

In the same paper we describe another algorithm which uses a completely different technique, a variant of the Buchberger-Möller algorithm

(see Section 6), based on linear algebra. It is well-suited to low degree hypersurfaces.

```
/**/ ImplicitHypersurface(P, [s^2, s*t, t^2], "Direct");
ideal(y^2 -x*z)
```

Both algorithms, in the case of rational coefficients, use modular methods computing the result for some primes, combining them using Chinese Remaindering, and reconstructing the rational coefficients of g using the fault-tolerant rational reconstruction described in Section 7.2.

§6. Ideals of Points, 0-Dimensional Schemes

Let X be a non-empty, finite set of points in K^n , then the set of all polynomials in $K[x_1, \dots, x_n]$ which vanish at all points in X is an ideal, I_X . One reason this ideal is interesting is because captures the “ambiguity” present in a polynomial function which has been interpolated from its values at the points of X . How best to compute a set of generators for I_X , or a Gröbner basis, knowing the points X ?

If X contains a single point (a_1, \dots, a_n) then we can write down immediately a Gröbner basis, namely $[x_1 - a_1, \dots, x_n - a_n]$. If X contains several points we could just intersect the ideals for each single point, and these intersections may be determined via Gröbner basis computations; while fully effective and mathematically elegant this approach is computationally disappointing.

A more efficient method is the Buchberger-Möller algorithm [14]. Somewhat astonishingly this uses just simple linear algebra to determine the Gröbner basis. The original algorithm was closely analysed in [6], then later generalized to zero-dimensional schemes [11], where it turned out that it also incorporates the well-known FGLM algorithm for “changing term-ordering” of a Gröbner basis.

The BM algorithm is also amenable to a modular approach, a technique for achieving particularly efficient computation with rational numbers. The CoCoA implementation uses this approach.

```
/**/ P ::= QQ[x,y];
/**/ points := Mat([[10, 0], [-10, 0], [0, 10], [0, -10],
                    [7, 7], [-7, -7], [7, -7], [-7, 7]]);
/**/ indent(IdealOfPoints(P, points));
ideal(
  x^2*y + (49/51)*y^3 + (-4900/51)*y,
  x^3 + (51/49)*x*y^2 -100*x,
  y^4 + (-2499/2)*x^2 + (-2699/2)*y^2 +124950,
  x*y^3 -49*x*y )
```

The simple use of linear algebra in the BM algorithm makes it a good candidate for identifying “almost-vanishing” polynomials for sets of *approximate* points: for instance, the points in the example above “almost lie on” a circle of radius 9.95 centred on the origin. The notion of Gröbner basis does not generalize well to an “approximate context” because the algebraic structure of a Gröbner basis is determined by Zariski-closed conditions (*i.e.* the structure is valid when certain polynomials vanish); instead the notion of a *Border Basis* is better suited since its structure is valid dependent on a Zariski-open condition (*i.e.* provided a certain polynomial *does not* vanish). So long as the approximate points are not too few nor too imprecise the BM algorithm can compute at least a partial Border Basis, and this should identify any “approximate polynomial conditions” which the points the almost satisfy (see Abbott, Fassino, Torrente [10]).

We can ask CoCoA to allow a certain approximation on the coordinates of the points:

```

/**/ epsilon := ColMat([0.1, 0.1]); // coord approximation 0.1
/**/ AV := TmpNBM(P, points, epsilon).AlmostVanishing; indent(AV);
[
  x^2 +(4999/5001)*y^2 -165000/1667, // almost a circle
  x*y^3 -49*x*y,
  y^5 -149*y^3 +4900*y
]

/**/ epsilon := ColMat([0.01, 0.01]); // coord approximation 0.01
/**/ AV := TmpNBM(P, points, epsilon).AlmostVanishing;
/**/ indent(AV); // with this epsilon, not near a conic
[
  x^2*y +(49/51)*y^3 +(-4900/51)*y,
  x^3 +(51/49)*x*y^2 -100*x,
  y^4 +(-2499/2)*x^2 +(-2699/2)*y^2 +124950,
  x*y^3 -49*x*y
]

```

§7. Gröbner bases and rational coefficients

It is well known that computations with coefficients in \mathbb{Q} can often be very costly in terms of both time and space. For Gröbner bases over \mathbb{Q} we are free to multiply the polynomials by any non-zero rational; so we can clear denominators and remove integer content. This does yield some benefit, but is not wholly satisfactory.

Sometimes the Gröbner basis has complicated coefficients (*i.e.* we mean *big numerators and denominators*), but more often the coefficients in the answer are reasonably sized, while the computation to obtain them involved far more complicated coefficients: this problem is known *intermediate coefficient swell*.

The phenomenon of coefficient swell is endemic in computer algebra, and many techniques have been investigated to tackle this problem. We illustrate what CoCoA offers.

7.1. TwinFloat

CoCoA offers floating-point arithmetic with a heuristic guarantee of correctness: the aim is to combine the speed of floating-point computation with the reliability of exact rational arithmetic — for a fuller description see the article [2]. Normally a twin-float computation will produce either a good approximation to the correct result or an indication of failure; strictly, there is a very small chance of getting a wrong result, but this never happens in practice.

To perform a computation with twin-floats the user must first specify the required precision; CoCoA will then check heuristically that the result of every computation has at least that precision. If the check fails then CoCoA signals an “insufficient precision” error; the user may then restart the computation specifying a higher precision. Although twin-float values are, by definition, approximate, all input values are assumed to be exact (so they can be converted to a twin-float of any precision).

It is also possible to reconvert a twin-float value to an exact rational number. Like all other twin-float operations, this conversion may fail (because of “insufficient precision”). Printing out a twin-float automatically checks if the value can be converted to a rational as rationals are easier to read and comprehend.

```

/**/ RR16 := NewRingTwinFloat(16);
/**/ use RR16_X ::= RR16[x,y,z];
/**/ f := 12345678*x+1/456789;
/**/ f;           // both coeffs are printed as rationals
12345678*x +1/456789
/**/ f * 10^3;    // first coeff is printed in "floating-point"
0.12345678*10^11*x +1000/456789
/**/ f * 10^5;    // both coeffs are printed in "floating-point"
0.12345678*10^13*x +0.2189194573

```

Twin-floats include a (heuristically guaranteed) test for zero; this means it is possible to compute Gröbner bases with twin-float coefficients. One reason for wanting to do this is that often the Gröbner

basis over the rationals involves “complicated fractions” (*i.e.* whose numerator and denominator have many digits), and arithmetic with such complicated values can quickly become very costly. In contrast, with twin-floats the arithmetic has fixed cost (dependent on the precision chosen, of course). These characteristics are exploited in CoCoA for the computation of *gin* described in Section 4.

7.2. (Fault-tolerant) Rational reconstruction

A widely used technique for avoiding intermediate coefficient swell is to perform the computation modulo one or more prime numbers, and then lift/reconstruct the final result over \mathbb{Q} . We call this the **modular approach**. There are two general classes of method: **Hensel Lifting** and **Chinese Remaindering**, the first is not universally applicable but does work well for polynomial gcd and factorization, while the second is widely applicable and works well in most other contexts.

The modular approach has been successfully used in numerous contexts, here are a few examples: polynomial factorization [20], determinant of integer matrices [9], ideals of points (see Section 6), and implicitization (see Section 5.2).

In any specific application there are two important aspects which must be addressed before a modular approach can be adopted, and there is no universal technique for addressing these issues.

- knowing how many different primes to consider to guarantee the result (*i.e.* find a realistic bound for the size of coefficients in the answer);
- handling *bad primes*: namely those whose related computation follows a different route, yielding an answer with the wrong “shape” (*i.e.* which is not simply the modular reduction of the correct non-modular result).

In the context of Gröbner bases we do not have good, general solutions to either of these issues. One of the first successes in applying modular techniques to Gröbner basis computation appeared in [12]. Finding good ways to employ a modular approach for Gröbner bases is still an active area. CoCoA does not currently use a modular approach for general Gröbner basis computations.

A vital complement to the modular computation is the reconstruction of the final, rational answer from the modular images. CoCoA offers functions for combining two residue-modulus pairs into a single “combined” residue-modulus pair (*i.e.* using the chinese remainder theorem). It also offers functions for determining a “simple” rational number corresponding to a residue-modulus pair; this is called **rational reconstruction**. Correct reconstruction can still be achieved even

in the presence of a few “faulty residues” (see [3]); this fault-tolerance was exploited in the functions for hypersurface implicitization (see Section 5.2).

Here we see how two modular images can be combined in CoCoA (using “CRTPoly” in this case), and then the correct rational result is reconstructed from the combined residue-modulus pair.

```

/**/ P1 := ZZ/(12347)[x];
/**/ P2 := ZZ/(23459)[x];
/**/ f1 := ReadExpr(P1, "x/1234-1/5"); f1; // in P1
-5293*x -4939
/**/ f2 := ReadExpr(P2, "x/1234-1/5"); f2; // in P2
-1806*x -4692

/**/ use P := QQ[x];
/**/ combined:=CRTPoly(-5293*x-4939, 12347, -1806*x-4692, 23459);
/**/ combined; // in P
record[modulus := 289648273, residue := 79571122*x +115859309]
/**/ RatReconstructPoly(combined.residue, combined.modulus);
(1/1234)*x -1/5

```

This reconstruction is exploited in CoCoA for the computation of the implicitization of hypersurfaces described in Section 5.2.

§8. Gröbner bases in C++ with CoCoALib

As mentioned in Section 1.2, our aim is to make computation using CoCoALib as easy as using CoCoA-5. To illustrate this, here is the first example from Section 2 but in C++:

```

ring P = NewPolyRing(RingQQ(), symbols("x,y,z"));
ideal I = ideal(ReadExpr(P, "x^3 + x*y^2 - 2*z"),
               ReadExpr(P, "x^2*y^3 - y*z^2"));
cout << GBasis(I);

```

In comparison to CoCoA-5, this C++ code is more cumbersome and involved, though we maintain that it is still reasonably comprehensible (once you know that `cout <<` is the C++ command for printing).

We have designed CoCoA-5 and CoCoALib together with the aim of making it easy to develop a prototype implementation in CoCoA-5, and then convert the code into C++. To facilitate this conversion we have, whenever possible, used the same function names in both CoCoA-5 and CoCoALib, and we have preferred traditional “functional” syntax in CoCoALib over object oriented “method dispatch” syntax (*e.g.* `GBasis(I)`

rather than `I.GBasis()`). This means that most of the CoCoA-5 examples given here require only minor changes to become equivalent C++ code for use with CoCoALib.

Naturally, a Gröbner basis is rarely computed in isolation; it is usually just one step in a higher-level computation. We have designed CoCoALib to be easy to use, *e.g.* for preparing the generators of the ideals whose Gröbner bases are to be computed, and also for further processing with the Gröbner bases after they have been computed. To maintain the “friendly” tradition of CoCoA software, our design of CoCoALib follows these aims:

- Designed to be **easy and natural to use**
- Execution speed is **good**
- Well-documented, including **many example programs**
- **Free and open source** C++ code (GPL3 licence)
- Source code is **clean and portable** (currently C++03)
- Design respects the underlying **mathematical** structures (inheritance, no templates)
- **Robust** (Motto: “No nasty surprises”), exception-safe, thread-safe

§9. Conclusion

The CoCoA software aims to make it easy for everyone to use Gröbner bases, whether directly or indirectly through some other function. The CoCoA-5 system is designed to be welcoming to those with little computer programming experience, while the CoCoALib library aims to make it easy for experienced programmers to use Gröbner bases in their own programs.

We hope this helps everyone to have their Gröbner basis!

References

- [1] J. Abbott, *The design of CoCoALib*, International Congress on Mathematical Software, 2006, pp. 205-215
- [2] J. Abbott, *Twin-Float Arithmetic*, Journal of Symbolic Computation, **47**, 2012, pp. 536-551.
- [3] J. Abbott, *Fault-tolerant modular reconstruction of rational numbers*, Journal of Symbolic Computation (in press).
- [4] J. Abbott, A.M. Bigatti, G. Lagorio, *CoCoA-5: a system for doing Computations in Commutative Algebra*. Available at <http://cocoa.dima.unige.it/>

- [5] J. Abbott and A.M. Bigatti, *CoCoALib: a C++ library for doing Computations in Commutative Algebra*. Available at <http://cocoa.dima.unige.it/cocoalib>
- [6] J. Abbott, A.M. Bigatti, M. Kreuzer, L. Robbiano, *Computing Ideals of Points*, Journal of Symbolic Computation **30**, 2000, 341–356.
- [7] J. Abbott, A.M. Bigatti, L. Robbiano, *Implicitization of Hypersurfaces*, arXiv preprint [arXiv:1602.03993](https://arxiv.org/abs/1602.03993).
- [8] J. Abbott, A.M. Bigatti, C. Söger, *Integration of libnormaliz in CoCoALib and CoCoA 5* Proceedings of ICMS 2014, LNCS Springer Verlag, 2014.
- [9] J. Abbott, M. Bronstein, T. Mulders, *Fast deterministic computation of determinants of dense matrices*, ISSAC '99 Proceedings, ACM New York, 1999, pp. 197–204
- [10] J. Abbott, C. Fassino, M.L. Torrente, *Stable border bases for ideals of points*, Journal of symbolic computation **43** (12), 2008, pp. 883–894
- [11] J. Abbott, M. Kreuzer, L. Robbiano, *Computing zero-dimensional schemes*, Journal of Symbolic Computation, **39**, 2005, pp. 31–49
- [12] E.A. Arnold *Modular Algorithms for Computing Groebner Bases*, Journal of Symbolic Computation, **35**, pp. 403–419 (2003).
- [13] A.M. Bigatti, R. La Scala, L. Robbiano, *Computing Toric Ideals*, Journal of Symbolic Computation, **27**, pp. 351–365 (1999).
- [14] B. Buchberger, H.M. Möller, *The construction of multivariate polynomials with preassigned zeros*, Proceedings of the European Computer Algebra Conference (EUROCAM '82), **144**, Lecture Notes in Comp.Sci., (1982), pp. 24–31.
- [15] M. Kreuzer, L. Robbiano, *Computational Commutative Algebra 1*, Springer Verlag, 2000.
- [16] Anders N. Jensen, *Gfan, a software system for Gröbner fans and tropical varieties* Available at <http://home.imf.au.dk/jensen/software/gfan/gfan.html>
- [17] M. Galassi et al, *GNU Scientific Library Reference Manual (3rd Ed.)*, ISBN 0954612078. *GSL*. Available at <http://www.gnu.org/software/gsl>
- [18] Teo Mora and Lorenzo Robbiano, *The Gröbner fan of an ideal*, J. Symb. Comput., **6** (2/3), 1988 pp. 183–208.
- [19] B. Sturmfels, *Gröbner bases and convex polytopes*, American Mathematical Soc., 1996.
- [20] H Zassenhaus, *On Hensel Factorization, I*, J. Number Theory vol. 1, pp. 291–311 (1969)

John Abbott
 Institut für Mathematik
 Universität Kassel ¹
 E-mail address: abbott@dimma.unige.it

¹J. Abbott is an INdAM-COFUND Marie Curie Fellow

Anna Maria Bigatti
Dipartimento di Matematica
Università degli Studi di Genova ²
E-mail address: `bigatti@dima.unige.it`

²This work was partly supported by the “National Group for Algebraic and Geometric Structures, and their Applications” (GNSAGA – INdAM)